

Copying clean text from Microsoft Word

Lonnie R. West – January 2006

Microsoft Word will replace standard keyboard characters with ones formatted to look better printed. Consider this sentence:

"I don't know if this'll work..." he replied. "Whenever I go there –even on a holiday– she won't speak to me."

Or a list with bullets, like this:

Groceries for today:

- 1) *Apples*
- 2) *Oranges*
- 3) *Vienna Sausage*

It is fine until you try to copy and paste the text into a web form or a plain old text editor; then you end up something really odd where the ellipsis, quotations marks, and dashes are replaced by indecipherable characters.

Fortunately, you can still use Word and take advantage of its editing and spell-checking power, and still export a clean ASCII text file suitable for pasting into web forms or other applications that do not use Word's extended fonts.

There are a couple extra steps you need to take (maybe someone out there has had the bright idea of making a plug-in for Word that gives you a "Copy Text As ASCII" option, but I haven't seen it yet!) to do this, but it will save you HOURS of work later on.

Make sure you have saved your final changes, and do the following with the document still open in Word:

- 1) Select **File->Save As** from the top menu,
- 2) In the "Save As Type" dropdown list at the bottom, select **Plain Text (*.txt)**,
- 3) Click **Save**,
- 4) You will be presented with a file conversion dialog box. Set the following options:
 - a. Select **MS-DOS** as the text encoding,
 - b. Put a checkmark next to the **Allow character substitution** option.
- 5) Click **OK**.
- 6) Close the document

You now have a clean ASCII text file. Just open it up in Word, or any other text editing program, and you can cut and paste at will, and everything will transfer correctly!

Lonnie West

www.zippybackflash.com

lonnie@zippybackflash.com

Adding a 'Copy As TXT' option to Microsoft Word

Lonnie R. West – January 2006

After much searching, I decided to make my own Macro in word that does the conversion for me. It works very simply: just highlight the text I want, and select Copy as TXT from the Edit menu. The results can be pasted into anything, and looks correct!

Interested? Here's how to do it!

Making the Macro

- Open Microsoft Word
- Select **File->Close** to close the current blank document.
- Select **Tools->Macro->Visual Basic Editor**
- The Microsoft Visual Basic editor will open.
- Double click on the "**ThisDocument**" template under "**Normal->Microsoft Word Objects**"
- Paste the code from the CopyAsTXT Code Box (next page) into the editing window.
- Select **File->Save Normal**
- Select **File->Close** (You will return to Microsoft Word)

Congratulations! You have now done the hard part. Now all we need to do is add this to our Edit menu, so you can access it whenever you want it.

Adding the Macro to the Edit menu

- If Word is still open, close it. We need a freshly opened copy for this next part.
- Open Microsoft Word and close the currently open document (if there is one open)
- Click **File->Open**
- Change File Type to **Template**
- Open the **Normal.dot** template (Usually in the folder:
C:\Documents and Settings\user name\Application Data\Microsoft\Templates)
- Select **Tools->Customize...**
- Click on the **Commands** tab
- Scroll the Categories down until you find **Macros**, click it
- Drag the **CopyAsTXT** item under **Commands**, and drop it in the Edit menu
(Edit menu will drop down when you drag it there, put it under the regular Copy menu item.)
- Right-click on your new entry, and look for the **Name:** property
- Change the Name to: **Copy As TXT**
- Click **Close** on the Customize dialog
- Click the **Save** icon on the toolbar (Looks like a Floppy Disk)
- Close Microsoft Word

That's it! Now when you open Word and edit a document, you will see the "Copy as TXT" entry in the Edit menu. Try it out by selecting the first few paragraphs on Page #1 (which is full of typical problem items) and clicking "Copy as TXT". Now go into a simple editor (Notepad, for example) or web-based text entry (Yahoo mail or a web blog, for example) and paste it. Looks nice, eh?

CopyAsTXT Code Box

```
'=====
Sub CopyAsTXT()
'*****
' CopyAsTXT Macro
' Macro created 2/3/2006 by Lonnie R. West
'-----
' Copies the selected text as clean ASCII suitable for
' pasting into Web documents, DOS applications, or simple
' text editors.
' www.zippybackflash.com
'*****

    Dim mPath As String
    Dim mNewDoc As Document
    Dim myRange As Range
    Dim mErrorCode As Integer

    On Error GoTo ErrHandler

    mErrorCode = 1
    '=====
    'Get Path of original document
    mPath = Application.ActiveDocument.Path & "\cache.txt"
    'Select highlighted text
    Selection.Copy
    'Close original

    mErrorCode = 2
    '=====
    'Create a new temp doc
    Set mNewDoc = Documents.Add
    mNewDoc.Select
    'Paste selection
    Selection.Paste

    mErrorCode = 3
    '=====
    'Save this as text file and close
    Call Application.ActiveDocument.SaveAs(mPath,wdFormatDOSText,,,False,,,,,,,True, wdCRLF, False)
    Application.ActiveDocument.Close

    mErrorCode = 4
    '=====
    'Open Text file
    Call Documents.Open(mPath,False,True,False,,,False,,,wdOpenFormatText, msoEncodingUSASCII)
    'Select text from this doc
    Set myRange = ActiveDocument.Content
    myRange.Select
    'Copy to clipboard
    Selection.Copy

    mErrorCode = 5
    '=====
    'Close doc
    Application.ActiveDocument.Close
    Exit Sub

ErrHandler:
    Select Case mErrorCode
        Case 1
            MsgBox ("You need to select text to copy!")
        Case 2
            MsgBox ("Unable to move selection to temporary document space!")
        Case 3
            MsgBox ("Unable to save converted text to cache file!")
        Case 4
            MsgBox ("Unable to open cache file!")
        Case 5
            MsgBox ("Unable to close cache file!")
    End Select

End Sub
'=====
```